第6章　モナドと命令型言語の意味

IO は効率のために Haskell の処理系で特別扱いされる組込みのモナドであるが、他の言語の副作用を模倣するためにユーザがモナドを定義することも可能である。モナドを用いる利点は、“計算”の意味が変わっても、モナドの標準的な関数 *unitM*, *bindM* のみを用いている部分は、変更する必要がないところである。

以下では、簡単な命令型プログラミング言語（つまり副作用を持つ言語）を定義し、モナドを利用してその意味を与えることにする。Haskell で意味を与えるということは、結局はラムダ計算で意味を与えることになる。具体的には命令型プログラミング言語から Haskell へのコンパイラを作成する。

簡単な言語からはじめて様々な特徴をもつ言語を定義していく。名前がないと不便なので、これらの対象言語を Util (**U**til: **T**iny **I**mperative **L**anguage)[1]と呼び、必要により、UtilErr, UtilST, UtilCont, ... などのようにバージョンを表す接尾語をつけることにする。いろいろな特徴を導入していくにつれ、その“計算”を表すモナドの定義が変わることになる。

実際のコンパイラにはフロントエンド、つまり _____や _____ __が必要である。字句解析や構文解析の原理は Haskell でも C 言語などの命令型言語で記述するときと変わりはない。再帰下降構文解析法（あるいは LR 構文解析法）などの方法を利用する。(ただし、再帰下降法で構文解析部を記述するときに、後述のようにモナドを利用することができる。)

しかし、ここではこれらフロントエンドの作り方は既知のものとして、構文木ができた状態から話をはじめることにする。

6.1　構文規則

Util の構文木のデータ構造として、次のような Haskell データ型を使用する。

```
1  type Decl = (String, Expr)
2  data Expr = Const Target            -- 定数（Target は後述）
3            | Var String              -- 変数
4            | If Expr Expr Expr       -- if 文
5            | While Expr Expr         -- while 文
6            | Begin [Expr]            -- ブロック
7            | Let [Decl] Expr         -- let 式（関数定義）
```

[1]PHP, GNU などの略語の由来も参照すること。

```
8             | Val Decl Expr          -- val 式（変数定義）
9             | Lambda String Expr    -- ラムダ式
10            | Delay Expr            -- delay 式（後述）
11            | App Expr Expr         -- 関数適用
12            deriving Show
```

つまり、式 (Expr) とは、定数 (Const) または、変数 (Var) または、 **if** 式 (If)、**let** 式 (Let)、ラムダ式 (Lambda)、関数適用 (App) などからなる。（あとから必要に応じて構文要素を追加することにする。）

具体的な構文としては次のような BNF で定義されていると仮定する。（演算子の優先順位なども適切に宣言されているとする。）

Expr → *Const* | *Var* | （ *Expr* ）
| **if** *Expr* **then** *Expr* **else** *Expr* | **while** *Expr* **do** *Expr*
| **begin** *Exprs* **end**
| **let** *Decls* **in** *Expr* | **val** *Decl* **in** *Expr*
| \ *Var* −> *Expr* | *Expr Expr*
| *Expr* + *Expr* | *Expr* ⋆ *Expr* | ··· （他の中置演算子）···

Exprs → *Expr* | *Expr* **;** *Exprs*

Decl → *Var* = *Expr*

Decls → *Decl* | *Decl* **;** *Decls*

ここに示されていないが、定数 *Const* と変数 *Var* の字句の定義は Haskell と同じとする。ただし、_（アンダーバー）から始まる変数名はコンパイラ内部で使用するために予約済みとする。

そして、次のような関数も既に定義されているものと仮定する。

```
myParse :: String -> Expr   -- 字句解析・構文解析の関数
```

- “**val** x=2⋆2 **in val** y=x⋆x **in** y⋆y”というソースプログラムは Expr 型のデータとして次のように構文解析される。

```
Val ("x", (App (App times (Const (TLit (Int 2))))
               (Const (TLit (Int 2)))))
    (Val ("y", (App (App times (Var "x")) (Var "x")))
               (App (App times (Var "y")) (Var "y"))))
```

  ただし、*times* は ⋆に対応する Expr の式である。

- “\ f -> \ x -> f x” という式は、

```
Lambda "f" (Lambda "x" (App (Var "f") (Var "x")))
```

  というデータに構文解析される。

- &&, ||は、それぞれ、

  b1 && b2 ↦ _____

  b1 || b2 ↦ _____

  という糖衣構文であるように構文解析されるようにしておく。

問 6.1.1 なぜ、&&や||はプリミティブ関数として定義すると良くないのか?

Target 型はコンパイラのターゲット言語である Haskell（のサブセット）を表現する型である。

```
1  type TDecl = (String, Target)
2  data Target = TLit Literal                -- 定数
3              | TVar String                 -- 変数
4              | TIf Target Target Target    -- if文
5              | Let [TDecl] Target          -- let式（変数・関数定義）
6              | TLambda1 String Target      -- ラムダ式
7              | TApp1 Target Target         -- 関数適用
8              | TUnitM Target
9              | TBindM Target Target
10            deriving (Show,Eq)
11 data Literal = Str String | Int Integer | Frac Rational
12             | Char Char                 deriving (Show,Eq)
```

Util のコンパイラとは次のような型を持つ関数である。

```
comp :: Expr -> Target   -- コンパイラ
```

6.2 抽象構文と具象構文

ところで、上の Util の構文規則は _____（ambiguous）である。通常は曖昧さを避けるために、

$$
\begin{array}{lcl}
\mathit{Expr} & \rightarrow & \mathit{Expr} + \mathit{Term} \mid \mathit{Term} \\
\mathit{Term} & \rightarrow & \mathit{Term} * \mathit{Factor} \mid \mathit{Factor} \\
\mathit{Factor} & \rightarrow & \mathit{Const} \mid (\ \mathit{Expr}\ )
\end{array}
$$

のように曖昧さを避けるために構文規則を工夫する。この後者のように実際に構文解析に用いるための構文を _____（concrete syntax）という。

それに対して、いったん構文解析が終了してしまえば、曖昧さを避けるための補助的な仕掛けは必要なくなり、本質的な構造のみを扱えばよい。そのため、前者のような構文規則で十分である。このように構成要素の本質的な関係を記述した構文のことを _____（abstract syntax）という。

この章で “構文” と呼んでいるのは、この抽象構文のことである。データ型 Expr の定義は、抽象構文を代数的データ型として直訳したものである。

6.3 コンパイラの定義

comp の定義は次のようになる。個々の構文要素に対する定義は比較的平易である。

```
1  comp :: Expr -> Target
2  comp (Const c)         = TUnitM c
3  comp (Var x)           = TUnitM (TVar x)
```

```
4   comp (Val (x, m) n)    = comp m `TBindM` TLambda1 x
5                             (comp n)
6   comp (Let decls n)     = TLet (map (\ (x, m) ->
7                                       let TUnitM c = comp m
8                                         in (PVar x, c)) decls)
9                                 (comp n)
10  comp (App f x)         = comp f `TBindM` TLambda1 "_f"
11                            (comp x `TBindM` TLambda1 "_x"
12                            (TApp1 (TVar "_f") (TVar "_x")))
13  comp (Lambda x m)      = TUnitM (TLambda1 x (comp m))
14  comp (Delay m)         = TUnitM (comp m)
15  comp (If e1 e2 e3)     = comp e1 `TBindM` TLambda1 "_b"
16                            (TIf (TVar "_b") (comp e2) (comp e3))
17  comp (While e1 e2)     = TLet [(PVar "_while",  body)]
18                                 (TVar "_while")
19     where body = comp e1 `TBindM` TLambda1 "_b"
20                  (TIf (TVar "_b")
21                       (comp e2 `TBindM` TLambda1 (TVar "_while"))
22                       (TUnitM (TVar "()")))
23  comp (Begin [e])       = comp e
24  comp (Begin (e:es))    = comp e `TBindM` TLambda1 (TVar "_")
25                            (comp (Begin es))
```

右辺で使われている _f, _x, _b, _while などの識別子は、Util ソースプログラム中に使われている識別子と衝突しないように選んでいる。

comp 関数を理解するために、変換前と変換後をそれぞれ Util と Haskell の文法で記述したのが次の表である。（詳細はソースプログラムを参照すること。）ただし bindM や unitM など、末尾に M が付く関数名は、対象のモナドに応じて適切な名前に置き換えられるものとする。

この表のなかで ソース中で *Italic* フォントで示されている *m*, *n* などは任意の Util の式で、ターゲット中で *m'*, *n'* のように ' （プライム）が付いている式は、その comp による変換後の Haskell の式を表す。なお、delay については内部的に使用されるので、実際には Util のソースプログラムに現れることはない。

ソース（Util）	ターゲット（Haskell）
c（ただし c は定数）	`unitM c`
x（ただし x は変数）	`unitM x`
val x = *m* **in** *n*	*m'* `` `bindM` `` `\ x -> n'`
let f = \ x −> *m* g = \ y −> *n* **in** *n*	**let** f = \ x -> *m'* x = \ y -> *n'* **in** *n'*
f x	*f'* `` `bindM` `` `\ _f ->` *x'* `` `bindM` `` `\ _x ->` `_f _x`
\ x −> *m*	`unitM (\ x -> m')`
delay *m*	`unitM m'`
if *c* **then** *t* **else** *e*	*c'* `` `bindM` `` `\ _b -> if _b then t' else e'`
while *c* **do** *t*	**let** _while = *c'* `` `bindM` `` `\ _b ->` **if** _b **then** *t'* `` `bindM` `` `\ _ ->` _while **else** () **in** _while
begin *s*; *t*; *u* **end**	*s'* `` `bindM` `` `\ _ ->` *t'* `` `bindM` `` `\ _ ->` *u'*

また、Util プログラム中の+, −, ⋆ などの二項演算子は、他の関数が `unitM (\ x -> ... )` という形に変換されること、戻り値はアクションを持たないことから、同名の Haskell 内のオペレータを用いて、それぞれ

```
unitM (\ x -> unitM (\ y -> unitM (x+y)))
unitM (\ x -> unitM (\ y -> unitM (x-y)))
unitM (\ x -> unitM (\ y -> unitM (x*y)))
```

という Haskell の式に置換するようにしておく。すると、comp 関数による他の部分の変換と整合する。

ソース（Util）	ターゲット（Haskell）
⊗（ただし ⊗ は二項演算子）	`unitM (\ x -> unitM (\ y -> unitM (x ⊗ y)))`

この comp を用いて、例えば次の Util プログラムを変換[2]すると

```
fact = \ n −> if n==0 then 1 else n*fact(n−1)
```

次のような Haskell のプログラムが得られる。

```
1  fact = \ n ->
2    ((unitM (\ x -> unitM (\ y -> unitM (x == y))) `bindM`
3        \ _f -> unitM n `bindM` \ _x -> _f _x)
4       `bindM` \ _f -> unitM 0 `bindM` \ _x -> _f _x)
5      `bindM`
```

[2]ただし、この fact 関数は副作用を含んでいないので、この変換自体にはあまり意味はない。

```
6       \ _b ->
7         if _b then unitM 1 else
8           (unitM (\ x -> unitM (\ y -> unitM (x * y))) `bindM`
9              \ _f -> unitM n `bindM` \ _x -> _f _x)
10           `bindM`
11           \ _f ->
12             (unitM fact `bindM`
13                \ _f ->
14                  ((unitM (\ x -> unitM (\ y -> unitM (x - y)))
15                     `bindM` \ _f -> unitM n
16                       `bindM` \ _x -> _f _x)
17                         `bindM` \ _f -> unitM 1
18                           `bindM` \ _x -> _f _x)
19                    `bindM` \ _x -> _f _x)
20                `bindM` \ _x -> _f _x
```

これは多くの冗長な部分を含んでいるので、前述の monad law などを利用して単純化すると、次のような式が得られる。

```
1   fact = \ n ->
2         if n == 0 then unitM 1 else
3              fact (n - 1) `bindM` \ _x ->
4              unitM (n * _x)
```

6.4 最初のバージョン – Util1

最初のバージョン Util1 では、モナドはトリビアルな計算（何もしない計算）としておく。つまり、Util1 は副作用を持たない言語である。

```
1   type I a = a
2
3   unitI :: a -> I a
4   unitI a = a
5
6   bindI :: I a -> (a -> I b) -> I b
7   m `bindI` k = k m
```

このとき、

```
let fact = \ n -> if n==0 then 1 else n*fact(n-1) in fact 9
```

という Util プログラムをコンパイルして実行すると、同じプログラムを Haskell として実行したときと同じ `362880` という値になる。

6.5 UtilST – 状態の導入

Util に更新（代入）可能な状態の概念を導入する。C 言語や Java 言語のように、変数に対して代入を導入することも可能であるが、非本質的な部分が多く

なってしまうので、ここで紹介する例では、2つだけ更新可能な“参照”xPとyPを導入することにする。2という数は別に本質的なものではなく、いくつにすることも可能である。

例えば、

```
begin setM xP 1; setM xP (getM xP+3); getM xP end
```

というUtilSTプログラムを評価すると、_という結果が得られる。

状態を導入するために、やはり“計算の型”を定義する必要がある。まず 次のSTを定義する。

```
1  type ST s a = s -> (a, s)
2
3  unitST :: a -> ST s a
4  unitST a = ___________________
5
6  bindST :: ST s a -> (a -> ST s b) -> ST s b
7  m `bindST` k = ____________________________________________
```

unitST aは状態(s)の変更を行なわず、aをそのまま返す計算である。m `bindST` kは、mで変更された状態(s1)をそのまま、kに受渡す計算である。

プリミティブは次のように定義する。

```
1  type Pos s a = s -> (a, a -> s)
2
3  xP :: Pos (x, y) x
4  xP = \ (x, y) -> (x, \ x1 -> (x1, y))
5
6  yP :: Pos (x, y) y
7  yP = \ (x, y) -> (y, \ y1 -> (x, y1))
8
9  getST :: Pos s a -> ST s a
10 getST p = \ s -> (fst (p s), s)
11
12 setST :: Pos s a -> a -> ST s ()
13 setST p v = \ s -> ((), snd (p s) v)
14
15 -- 例えば getST xP ≡ \ (x,y) -> (x (x,y))
```

setSTは状態を書き換え、またgetSTは状態の値の一部を複製している。

UtilSTのsetM, getM, ... という関数は、そのままHaskellのsetM, getM, ... にコンパイルされるようにしておく[3]。

[3]このプリント中では、getMのように“M”という接尾辞を使っているときは、一般のモナドに適用可能な関数や型であり、getSTのように“M”以外の(例えば“ST”)という接尾辞を使っているときは、特定のモナド STに関する関数や型であるものと約束する。つまり、Mという接尾辞がついている識別子は実際に使うときに適宜名前を付け替える。

ソース（Util）	ターゲット（Haskell）
`setM p m`	`p' `bindM` \ _p -> m' `bindM` \ _x -> setM _p _x`
`getM p'`	`p' `bindM` \ _p -> getM _p`

UtilST プログラム（右は対応する C プログラム）：

```
1  fact = \ n ->
2    begin
3      setM xP 1; setM yP n;
4      while getM yP > 0 do begin
5        setM xP (getM xP * getM yP);
6        setM yP (getM yP - 1)
7      end;
8      getM xP
9    end
```

```
1  int fact(int y) {
2
3    int x = 1;
4    while (y > 0) {
5      x = x * y;
6      y = y - 1;
7    }
8    return x;
9  }
```

をコンパイルすると次のような Haskell の関数（一部、見易くするために変数名の変更などをしている）になり、

```
1  fact n = setST xP 1 `bindST` \ _ ->
2           setST yP n `bindST` \ _ ->
3           (let _while = getST yP `bindST` \ y ->
4                         if y > 0 then
5                              getST xP          `bindST` \ x ->
6                              getST yP          `bindST` \ y ->
7                              setST xP (x*y)    `bindST` \ _ ->
8                              getST yP          `bindST` \ y ->
9                              setST yP (y-1)    `bindST` \ _ ->
10                             _while
11                           else unitST ()
12              in _while)    `bindST` \ _ ->
13          getST xP
```

`fact 9` を実行する（`fst (fact 9 (0, 0))`）とその結果は `362880` になる。

階乗の場合、普通に関数的な定義を書いた方が簡潔だが、パラメータの数が多い場合などは、このような命令的な書き方の方が簡潔になる場合もありうる。

この ST の定義では、エラー処理を考慮していない。エラー処理を行なうためには、この ST と（後述する）Maybe の定義を合成する必要がある。参考までに、次のようなモナドになる。

```
1  type EST s a = s -> Maybe (a, s)
2
3  unitEST :: a -> EST s a
4  unitEST a = \ s -> unitE (a, s)
5
6  bindEST :: EST s a -> (a -> EST s b) -> EST s b
7  m `bindEST` k = \ s0 -> case m s0 of
8                             Just (a, s1) -> k a s1
9                             Nothing      -> Nothing
```

6.6 UtilIO – 入出力の導入

入出力は、入出力ストリームを状態の一種と考えれば、6.5 節の UtilST と同じ方法で取り扱うことができる。

計算のモナドの定義は 6.5 節と基本的に同じだが、状態に入力と出力のストリームを表す String 型の部分を追加しておく。

```
1  type MyState s = (s, String, String)
2  type MyIO s a = ST (MyState s) a
3        -- つまり、 MyState s -> (a, MyState s)
```

入出力のプリミティブの定義は次のようになる。

```
1  getIO :: Pos s a ->MyIO s a
2  getIO p = \ (s, i, o) -> (fst (p s), (s, i, o))
3
4  setIO :: Pos s a -> a -> MyIO s ()
5  setIO p v = \ (s, i, o) -> ((), (snd (p s) v, i, o))
6
7  readIO :: () -> MyIO s Char
8  readIO () = ____________________________________________
9
10 writeIO :: Show s => s -> MyIO ()
11 writeIO v  = ____________________________________________________
12
13 eofIO :: () -> MyIO s Bool
14 eofIO () = \ (s, i, o) -> (null i, (s, i, o))
```

readIO は入力ストリームから 1 文字を取出し、writeIO v は出力ストリーム o に v を String に変換したものを追加している[4]。

すると、次の UtilIO プログラム

```
1  let sq = \ x -> if x>0 then x*x else 0-x*x
2    in writeM (sq 2)
```

を実行したときの出力は、"4"になる。

6.7 UtilE – エラー処理の導入

次に Util にエラー処理を導入する。UtilErr は、生真面目に (?) エラー処理を行ない、部分式にエラーがあれば式全体もエラーになるようにする。この場合、

[4] ++の計算量は左オペランドの長さに比例するので、この定義のように文字列の後ろに新しい文字列を追加（++）していくと、出力文字列が長くなるにしたがって効率が悪くなる。これを避けて効率の良い定義を与えることも可能であるが、ここでは簡単のために++を使った定義を採用する。

UtilErr は（Haskell のような）遅延評価ではなくて、関数の引数を必ず先に評価する _____（eager evaluation）をシミュレートすることに注意する必要がある。

エラーと正常な振舞いを区別するために、次のようにデータ型 Maybe を使用する。

```
-- Prelude に定義済み
data Maybe a = _________ | ________ deriving (Show,Eq)
```

正常な振舞いは Just という構成子で表す。エラーの場合は Nothing という構成子を用いる。この型に対して次のような補助関数を定義しておく。

```
1  unitE :: a -> Maybe a
2  unitE a = Just a
3
4  bindE :: Maybe a -> (a -> Maybe b) -> Maybe b
5  (Just a) `bindE` k = ____
6  Nothing  `bindE` k = _________
```

m `bindE` k は、まず m を計算し、その計算が正常終了すれば、その値を k という関数に渡す。しかし、いったん m でエラーが起こると、k は評価されず、_____ことを表している。

さらに、補助関数を定義しておく。failE は _____ときに用いる UtilErr の計算の型に特有の関数である。

```
1  failE :: String -> Maybe a
2  failE _ = Nothing
```

“プリミティブ関数” もエラー処理を利用するように書き換えることができる。例えば、割り算（/）は次のような Haskell の式に置換されるようにする。

```
\ x -> unitM (\ y -> if y==0 then failM "Division by 0"
                             else unitM (x/y))
```

これで 0 で割ろうとした場合にはエラーが報告される。

例えば “(\ x -> 0) (1/0)” のような式は、Haskell では _____が、UtilErr では次のような Haskell プログラムに翻訳され、

```
(if 0 == 0 then failE "Division by 0"
           else unitE (1/0))          `bindE` \ _x ->
unitE 0
```

実行すると _____という結果になる。

6.8 例外処理の導入

例外のモナド E を利用して、Java の try〜catch のように例外を捕捉する構文を導入することも可能である。

BNF には以下の構文を追加する。

Expr → ... | **try** *Expr* **catch** *Expr*

“**try** *m* **catch** *h*” は *m* を評価し、エラーがなかった場合は、その戻り値を **try** 式の戻り値とする。しかし *m* の評価中にエラーが生じた場合は、*h* を評価する。“**try** *m* **catch** *h*”は“tryM (**delay** *m*) (**delay** *h*)”という式として構文解析されるようにしておく。

また、failM という Util の関数は、そのまま Haskell の failM にコンパイルされるようにしておく。この関数は、Java の throw 文に対応する。

ソース（Util）	ターゲット（Haskell）
try *m* **catch** *n*	tryM (delay *m'*) (delay *n'*)
failM m	*m'* `bindM` \ _x -> failM _x

tryE は _____ 関数である。

```
1  tryE :: Maybe a -> Maybe a -> Maybe a
2  tryE (Just v) h = ________
3  tryE Nothing h  = _
```

例えば

```
try 1/0 catch 99999
```

という UtilErr プログラムをコンパイルすれば、出力される Haskell プログラムは、

```
1  tryE (if 0 == 0 then failE "Division by 0" else unitE (1/0))
2       (unitE 99999)
```

であり、その結果は Just 99999.0 である。

6.9 UtilNonDet – 非決定性の導入

ここからは対象言語 Util に非決定性を導入する。非決定性（nondeterminism）とは _____ を言う。ある選択肢を選んだ結果、計算が失敗する場合がある、その場合は前の選択肢に戻って計算をやり直す（バックトラック）。非決定性は探索型のパズル・ゲームや構文解析プログラムなどで利用できる。バックトラックをプリミティブな機能として提供する言語としては、論理型言語の _____ が有名である。

非決定性の “計算” のモナドは通常のリストを使用する。

```
1  unitL :: a -> [a]
2  unitL a = [a]
3
4  appendL :: [a] -> [a] -> [a]
5  (x:xs) `appendL` ys = x : (xs `appendL` ys)
```

```
6   []     `appendL` ys = ys
7
8   bindL :: [a] -> (a -> [b]) -> [b]
9   (a:es) `bindL` k = k a `appendL` (es `bindL` k)
10  []     `bindL` k = []
```

このモナドは複数の選択肢を単にリストとして表現している。

実は unitL と bindL は、リストの内包表記を説明する時に使った unit :: a -> [a] と bind :: [a] -> (a -> [b]) -> [b] とまったく同一の関数である。

計算の失敗は空リストで表される。

```
1   failL :: String -> [a]
2   failL _ = ___
```

UtilNonDet は、構文は UtilErr と同じである。つまり、以下の構文を持つ。

$$Expr \quad \rightarrow \quad \ldots \mid \texttt{try}\ Expr\ \texttt{catch}\ Expr$$

try m catch h は、m を評価し、h は“バックトラック”が起こったときに評価される。

tryL は appendL そのものである。

```
1   tryL :: [a] -> [a] -> [a]
2   tryL = appendL
```

UtilNonDet プログラム

```
test0 = (try 1 catch 2) * (try 3 catch 4)
```

をコンパイルすると、次の Haskell プログラムが得られる。

```
1   test0 = tryL (unitL 1) (unitL 2) `bindL` \ x ->
2           tryL (unitL 3) (unitL 4) `bindL` \ y ->
3           unitL (x*y)
```

この、test0 は [x*y | x <- [1,2], y <- [3,4]] というリスト内包表記に対応する。test0 は、[3,4,6,8] となる。

また、次の UtilNonDet プログラム

```
test1 = (try 1 catch 2) / (try 0 catch 4)
```

をコンパイルすると、次の Haskell プログラムが得られる。

```
1   test1 = tryL (unitL 1) (unitL 2) `bindL` \ x ->
2           tryL (unitL 0) (unitL 4) `bindL` \ y ->
3           if y == 0 then failL "Division␣by␣0" else unitL (x/y)
```

test1 は、[0.25,0.5] となる。失敗している計算については結果に現れていないことに注意する。同じプログラムを UtilErr でコンパイルすると、UtilErr ではバックトラッキングが起こらないので、この計算は全体が失敗に終わる。

なお、次の headL を用いてリストの頭部を取ることにより、成功した最初の計算だけを返すことも可能である。

```
1  headL :: [a] -> a
2  headL (x:_) = x
```

headL test1 の値は 0.25 となる。この場合、Haskell が _____を採用しているため、他の選択肢の計算は行なわれない。そのため選択肢が無限個あるような場合でも最初の選択肢の計算結果を出力することができる。

問 6.9.1 非決定性と状態の両方の特徴を持つ計算の型として、

```
1  type STL s a =  s -> ([a], s)
2  type LST s a =  s -> [(a, s)]
```

の 2 つのバリエーションが考えられる。このそれぞれに対して、インタプリタの定義を完成させ、2 つの違いを説明せよ。

6.10 さらに詳しく知りたい人のために ...

Parsec [1] はモナドを利用した有名なパーサーライブラリーである。[3] にも、モナドを用いてパーサを構築する技法の解説がある。[2] はモナドを用いてインタプリタを構築する方法を解説している。[4] は、Prolog のカット等のオペレータの意味を整理している。

この章の参考文献

[1] Daan Leijen and Erik Meijer「Parsec: Direct Style Monadic Parser Combinators for the Real World」
Technical Report UU-CS-2001-35, Dept. of Comp. Sci, Universiteit Utrecht, 2001 年, http://www.cs.uu.nl/people/daan/parsec.html

[2] Philip Wadler「The essence of functional programming」
19th Annual Symposium on Principles of Programming Languages (invited talk), 1992 年 1 月

[3] Philip Wadler「Monads for functional programming」
Program Design Calculi, Proceedings of the Marktoberdorf Summer School, 1992 年 7–8 月

[4] Ralf Hinze「Prological Features in a Functional Setting Axioms and Implementations」
Third Fuji International Symposium on Functional and Logic Programming, 1998 年